

F08ATFP (PZUNGQR)

NAG Parallel Library Routine Document

Note: Before using this routine, please read the Users' Note for your implementation to check for implementation-dependent details. You are advised to enclose any calls to NAG Parallel Library routines between calls to Z01AAFP and Z01ABFP.

1 Description

This routine is intended to be used after a call to F08ASFP (PZGEQRF), which performs a QR factorization of an m by r complex matrix A_s . F08ASFP represents the m by m unitary matrix Q as a product of elementary reflectors.

F08ATFP (PZUNGQR) may be used to generate Q explicitly as a square matrix, or to form the n leading columns of Q , where $n \leq m$.

Alternatively, this routine may be called to compute the unitary matrix Q of the QR factorization of the k leading columns of the matrix A_s for $k \leq r$.

This routine returns the matrix Q in array A which on entry contained the details of elementary reflectors computed by F08ASFP (PZGEQRF). The position of the (1, 1) element of the matrix Q in the array A is given by (i_A, j_A) , which originally identified the (1, 1) element of the matrix A_s in the factorization routine F08ASFP (PZGEQRF). The distribution of the matrix Q conforms to the details in the description array IDESCA.

2 Specification

```

SUBROUTINE F08ATFP(M, N, K, A, IA, JA, IDESCA, TAU, WORK, LWORK,
1              INFO)
ENTRY        PZUNGQR(M, N, K, A, IA, JA, IDESCA, TAU, WORK, LWORK,
1              INFO)
COMPLEX*16   A(*), TAU(*), WORK(LWORK)
INTEGER      M, N, K, IA, JA, IDESCA(9), LWORK, INFO

```

The ENTRY statement enables the routine to be called by its ScaLAPACK name.

3 Data Distribution

3.1 Definitions

The following definitions are used in describing the data distribution within this document:

- m_p – the number of rows in the logical processor grid.
- n_p – the number of columns in the logical processor grid.
- p_r – the row grid coordinate of the calling processor.
- p_c – the column grid coordinate of the calling processor.
- M_b^X – the blocking factor for the distribution of the rows of a matrix X .
- N_b^X – the blocking factor for the distribution of the columns of a matrix X .
- $\text{numroc}(\alpha, b_\ell, q, s, k)$ – a function which gives the **number of rows or columns** of a distributed matrix owned by the processor with the row or column coordinate q (p_r or p_c), where α is the total number of rows or columns of the matrix, b_ℓ is the blocking factor used (M_b^X or N_b^X), s is the row or column coordinate of the processor that possesses the first row or column of the distributed matrix and k is either n_p or m_p . The Library provides the function Z01CAFP (NUMROC) for the evaluation of this function..
- $\text{indxg2p}(i_g, b_\ell, q, s, k)$ – a function which gives the processor row or column coordinate which possess the row or column index i_g of the distributed full matrix A . The arguments b_ℓ, q, s and k have the same meaning as in the function numroc. The Library provides the function Z01CDFP (INDXG2P) for the evaluation of this function.

3.2 Global and Local Arguments

The input arguments M, N, K, IA, JA and the array elements IDESCA(1) and IDESCA(3),...,IDESCA(8) are all global and so must have the same value on entry to the routine on every processor. The output argument INFO is global and so will have the same value on exit from the routine on each processor. The remaining arguments are local.

3.3 Distribution Strategy

On entry to this routine, the input values of M, A, IA, JA, IDESCA and TAU must be identically equal to the output values of the corresponding arguments on exit from the *QR* factorization routine F08ASFP (PZGEQRF).

In F08ASFP (PZGEQRF), the matrix Q is represented as a set of elementary reflectors but in F08ATFP (PZUNGQR), the matrix Q is explicitly computed. In both these routines, the details of Q are stored in local arrays denoted by A but the minimal storage requirements (as specified by the second dimensions) of A on a particular processor are, in general, different. In particular, if $r < n$ then F08ATFP (PZUNGQR) requires larger local arrays for A than in F08ASFP (PZGEQRF).

4 Arguments

- 1:** M — INTEGER *Global Input*
On entry: m , the number of rows of the Q . This should be identical to the number of rows of the matrix A_s as supplied to the *QR* factorization routine F08ASFP (PZGEQRF).
Constraint: $0 \leq M \leq \text{IDESCA}(3)$.
- 2:** N — INTEGER *Global Input*
On entry: n , the number of columns of the matrix Q that are required.
Constraints: $0 \leq N \leq \text{IDESCA}(4)$; $N \leq M$.
- 3:** K — INTEGER *Global Input*
On entry: k , the number of columns of the matrix A_s which should be used in the computation of the matrix Q . This index also identifies the number of elementary reflectors whose product defines Q .
Constraint: $0 \leq K \leq N$.
- 4:** A(*) — COMPLEX*16 array *Local Input/Local Output*
Note: array A is formally defined as a vector. However, you may find it more convenient to consider A as a 2-d array of dimension $(\text{IDESCA}(9), \gamma)$, where $\gamma \geq \text{numroc}(\text{JA} + \max(N, K) - 1, \text{IDESCA}(6), p_c, \text{IDESCA}(8), n_p)$.
On entry: details of the vectors which define the elementary reflectors as returned by F08ASFP (PZGEQRF).
On exit: the local parts of the first n columns of the m by m matrix Q . The distribution of the matrix Q is defined by the indices IA and JA and the description array IDESCA.
- 5:** IA — INTEGER *Global Input*
On entry: the row index of matrix A, i_A , that identifies the first row of Q to be generated.
Constraints: $1 \leq \text{IA} \leq \text{IDESCA}(3) - M + 1$.
- 6:** JA — INTEGER *Global Input*
On entry: the column index of matrix A, j_A , that identifies the first column of Q to be generated.
Constraints: $1 \leq \text{JA} \leq \text{IDESCA}(4) - \max(N, K) + 1$.

7: IDESCA(9) — INTEGER array *Local Input*

Distribution: the array elements IDESCA(1) and IDESCA(3),...,IDESCA(8) must be global to the processor grid and the elements IDESCA(2) and IDESCA(9) are local to each processor.

On entry: the description array for the matrix A . This array must contain details of the distribution of the matrix A and the logical processor grid.

IDESCA(1), the descriptor type. For this routine, which uses a cyclic 2-d block distribution, IDESCA(1) = 1;

IDESCA(2), the BLACS context (ICNTXT) for the processor grid, usually returned by Z01AAFP;

IDESCA(3), the number of rows, m_A , of the matrix A ;

IDESCA(4), the number of columns, n_A , of the matrix A ;

IDESCA(5), the blocking factor, M_b^A , used to distribute the rows of the matrix A ;

IDESCA(6), the blocking factor, N_b^A , used to distribute the columns of the matrix A ;

IDESCA(7), the processor row index over which the first row of the matrix A is distributed;

IDESCA(8), the processor column index over which the first column of the matrix A is distributed;

IDESCA(9), the leading dimension of the conceptual 2-d array A .

Constraints:

IDESCA(1) = 1;

IDESCA(3) \geq 0; IDESCA(4) \geq 0; IDESCA(5) \geq 1; IDESCA(6) \geq 1;

0 \leq IDESCA(7) \leq $m_p - 1$; 0 \leq IDESCA(8) \leq $n_p - 1$;

IDESCA(9) \geq max(1, numroc(IDESCA(3), IDESCA(5), p_r , IDESCA(7), m_p)).

8: TAU(*) — COMPLEX*16 array *Local Input*

Note: the dimension of the array TAU must be at least α , where $\alpha = \text{numroc}(\text{JA} + \text{K} - 1, \text{IDESCA}(6), p_c, \text{IDESCA}(8), n_p)$.

On entry: details of the elementary reflectors, as returned by a call to F08ASFP (PZGEQRF).

9: WORK(LWORK) — COMPLEX*16 array *Local Workspace/Local Output*

On exit: WORK(1) returns the minimum required value of LWORK.

10: LWORK — INTEGER *Local Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08ATFP (PZUNGQR) is called.

Constraint: LWORK \geq IDESCA(6) \times ($c_1 + c_2 + \text{IDESCA}(6)$), where

$c_1 = \text{numroc}(M + d_1, \text{IDESCA}(5), p_r, e_1, m_p)$;

$c_2 = \text{numroc}(N + d_2, \text{IDESCA}(6), p_c, e_2, n_p)$;

$d_1 = \text{mod}(\text{IA} - 1, \text{IDESCA}(5))$;

$d_2 = \text{mod}(\text{JA} - 1, \text{IDESCA}(6))$;

$e_1 = \text{indxc}2p(\text{IA}, \text{IDESCA}(5), p_r, \text{IDESCA}(7), m_p)$;

$e_2 = \text{indxc}2p(\text{JA}, \text{IDESCA}(6), p_c, \text{IDESCA}(8), n_p)$.

This value of LWORK can be calculated by using the Library function Z01CBFP; i.e.,

LWORK = Z01CBFP(M, N, IA, JA, IDESCA)

11: INFO — INTEGER *Global Output*

On exit: INFO = 0 unless the routine detects an error (see Section 5).

5 Errors and Warnings

If $\text{INFO} < 0$ an explanatory message is output and control returned to the calling program.

$\text{INFO} < 0$

On entry, one of the arguments was invalid:

if the k th argument is a scalar $\text{INFO} = -k$;

if the k th argument is an array and the j th element is invalid, $\text{INFO} = -(100 \times k + j)$.

This error occurred either because a global argument did not have the same value on all logical processors, or because its value on one or more processors was incorrect. An explanatory message distinguishes between these two cases.

6 Further Comments

Often Q is determined from the QR factorization of an m by r matrix A with $m \geq r$. The matrix Q may be computed by calling:

```
CALL F08ATFP (M,M,r,A,IA,JA,IDESCA,TAU,WORK,LWORK,INFO)
```

The leading r columns of Q may be obtained by:

```
CALL F08ATFP (M,r,r,A,IA,JA,IDESCA,TAU,WORK,LWORK,INFO)
```

The columns of Q returned by the last call form an orthonormal basis for the space spanned by the columns of A ; thus F08ASFP (PZGEQRF) followed by F08ATFP (PZUNGQR) can be used to orthogonalise the columns of A .

6.1 Algorithmic Detail

See [1] for details of the block method used by the routine.

6.2 Parallelism Detail

The Level 3 BLAS operations are carried out in parallel within the routine.

6.3 Accuracy

The computed matrix Q differs from an exactly unitary matrix by a matrix E such that

$$\|E\|_2 \leq \epsilon p(m, n),$$

where ϵ is the *machine precision*, $p(m, n)$ is a modest function of m and n .

7 References

- [1] Anderson E, Bai Z, Bischof C, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A, Ostrouchov S and Sorensen D (1995) *LAPACK Users' Guide* (2nd Edition) SIAM, Philadelphia
- [2] Golub G H and Van Loan C F (1989) *Matrix Computations* Johns Hopkins University Press (2nd Edition), Baltimore

8 Example

To form the leading four columns of the unitary matrix Q from the QR factorization of the matrix A , where

$$A = \begin{pmatrix} 0.96 - 0.81i & -0.03 + 0.96i & -0.91 + 2.06i & -0.05 + 0.41i \\ -0.98 + 1.98i & -1.20 + 0.19i & -0.66 + 0.42i & -0.81 + 0.56i \\ 0.62 - 0.46i & 1.01 + 0.02i & 0.63 - 0.17i & -1.11 + 0.60i \\ -0.37 + 0.38i & 0.19 - 0.54i & -0.98 - 0.36i & 0.22 - 0.20i \\ 0.83 + 0.51i & 0.20 + 0.01i & -0.17 - 0.46i & 1.47 + 1.59i \\ 1.08 - 0.28i & 0.20 - 0.12i & -0.07 + 1.23i & 0.26 + 0.26i \end{pmatrix}$$

The columns of Q form an orthonormal basis for the space spanned by the columns of A . The example uses a 2 by 2 logical processor grid and a 1 by 1 block for A .

Note: the listing of the Example Program presented below does not give a full pathname for the data file being opened, but in general the user must give the full pathname in this and any other OPEN statement.

8.1 Example Text

```
*      F08ATFP Example Program Text
*      NAG Parallel Library Release 2. NAG Copyright 1996.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER         DT
PARAMETER       (DT=1)
INTEGER         MB, NB, LW
PARAMETER       (MB=2,NB=MB,LW=100)
INTEGER         MMAX, NMAX, LDA, IAROW, IACOL
PARAMETER       (MMAX=6,NMAX=4,LDA=MMAX,IAROW=0,IACOL=0)
*      .. Local Scalars ..
INTEGER         IA, ICNTXT, IFAIL, INFO, JA, LWORK, M, MP, N, NP
LOGICAL         ROOT
CHARACTER*80    FORMAT
*      .. Local Arrays ..
COMPLEX*16     A(LDA,NMAX), TAU(MMAX), WORK(LW)
INTEGER         IDESCA(9)
*      .. External Functions ..
INTEGER        Z01CBFP
LOGICAL        Z01ACFP
EXTERNAL       Z01CBFP, Z01ACFP
*      .. External Subroutines ..
EXTERNAL       F08ASFP, F08ATFP, X04BRFP, X04BSFP, Z01AAFP,
+             Z01ABFP
*      .. Intrinsic Functions ..
INTRINSIC      MIN
*      .. Executable Statements ..
ROOT = Z01ACFP()
IF (ROOT) WRITE (NOUT,*) 'F08ATFP Example Program Results'
*
MP = 2
NP = 2
IFAIL = 0
*
CALL Z01AAFP(ICNTXT,MP,NP,IFAIL)
*
OPEN (NIN,FILE='f08atfpe.d')
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) M, N
```

```

      READ (NIN,*) FORMAT
*
      IF (M.LE.MMAX .AND. N.LE.NMAX) THEN
*
*       Set array descriptor for A, and read A from data file
*
      IDESCA(1) = DT
      IDESCA(2) = ICNTXT
      IDESCA(3) = M
      IDESCA(4) = N
      IDESCA(5) = MB
      IDESCA(6) = NB
      IDESCA(7) = IAROW
      IDESCA(8) = IACOL
      IDESCA(9) = LDA
*
      IFAIL = 0
      CALL X04BRFP(NIN,M,N,A,1,1,IDESCA,IFAIL)
*
      IA = 1
      JA = 1
      LWORK = MIN(LW,Z01CBFP(M,N,IA,JA,IDESCA))
*
      CALL F08ASFP(M,N,A,IA,JA,IDESCA,TAU,WORK,LWORK,INFO)
      IF (INFO.EQ.0) THEN
*
*       Generate Q
*
      CALL F08ATFP(M,N,N,A,IA,JA,IDESCA,TAU,WORK,LWORK,INFO)
*
      IF (INFO.EQ.0) THEN
        IF (ROOT) THEN
          WRITE (NOUT,*)
          WRITE (NOUT,99999) N
          WRITE (NOUT,*)
        END IF
        IFAIL = 0
*
        CALL X04BSFP(NOUT,M,N,A,IA,JA,IDESCA,FORMAT,WORK,IFAIL)
*
      END IF
    END IF
*
  END IF
*
  CLOSE (NIN)
*
  IFAIL = 0
  CALL Z01ABFP(ICNTXT,'N',IFAIL)
*
  STOP
*
99999 FORMAT (' The leading ',I2,' columns of Q')
      END

```

8.2 Example Data

F08ATFP Example Program Data

```

6 4                                     :Values of M and N
'F7.4'                                  :Value of FORMAT
( 0.96,-0.81) (-0.03, 0.96) (-0.91, 2.06) (-0.05, 0.41)
(-0.98, 1.98) (-1.20, 0.19) (-0.66, 0.42) (-0.81, 0.56)
( 0.62,-0.46) ( 1.01, 0.02) ( 0.63,-0.17) (-1.11, 0.60)
(-0.37, 0.38) ( 0.19,-0.54) (-0.98,-0.36) ( 0.22,-0.20)
( 0.83, 0.51) ( 0.20, 0.01) (-0.17,-0.46) ( 1.47, 1.59)
( 1.08,-0.28) ( 0.20,-0.12) (-0.07, 1.23) ( 0.26, 0.26) :End of matrix A

```

8.3 Example Results

F08ATFP Example Program Results

The leading 4 columns of Q

```

(-0.3110, 0.2624) (-0.3175, 0.4835) ( 0.4966,-0.2997) (-0.0072,-0.3718)
( 0.3175,-0.6414) (-0.2062, 0.1577) (-0.0793,-0.3094) (-0.0282,-0.1491)
(-0.2008, 0.1490) ( 0.4892,-0.0900) ( 0.0357,-0.0219) ( 0.5625,-0.0710)
( 0.1199,-0.1231) ( 0.2566,-0.3055) ( 0.4489,-0.2141) (-0.1651, 0.1800)
(-0.2689,-0.1652) ( 0.1697,-0.2491) (-0.0496, 0.1158) (-0.4885,-0.4540)
(-0.3499, 0.0907) (-0.0491,-0.3133) (-0.1256,-0.5300) ( 0.1039, 0.0450)

```
